
  
Abstract—High temperature hydrolysis is one of important steps 

to produce lignocellulosic ethanol using chemical process. Since the 
heating of the reactant needs considerable time, during the heating, 
the reaction has started. As a result, the hydrolysis usually runs at 
non-isothermal condition. This research explores the use of Monte 
Carlo simulation to deeply study the progress of non-isothermal 
hydrolysis by calculating the chain length distribution of the 
polysaccharide at various times. The accuracy of the Monte Carlo 
simulation was tested by comparing the results with the ones 
obtained by kinetics model via simultaneous differential equations 
for hydrolysis of relatively short chain polymers. It turned out that 
Monte Carlo simulation results are in good agreement with the ones 
of the kinetics models. It is observed also that Monte Carlo 
simulation can be easily applied for long chain polysaccharide, while 
the kinetics model is not feasible. 

I. INTRODUCTION 
TUDIES on lignocellulosic ethanol, one of renewable 
energy source alternatives for liquid transport fuels, are 

developing rapidity, such as study on the potential 
lignocellulosic raw material [1], [2], the efficient process 
condition [3]–[5], and the economic process condition [6], [7]. 
Likewise, studies on the homogeneous kinetic models of 
dilute-acid hydrolysis [8] as well as heterogeneous one [9], 
and enzymatic hydrolysis [10] have been conducted at various 
conditions. Commonly, the kinetic approaches describe the 
yield of monosaccharide, which represents the ratio of 
monosaccharide formed to the theoretical monosaccharide can 
be formed, for examples: investigation of the mechanism of 
acidic hydrolysis of cellulose [11]; kinetics of enzymatic 
hydrolysis of cellulose [10]; modeling using Monte Carlo 
simulation method of enzymatic hydrolysis of amilose and 
amylopectin of wet milled waxy, corn hybrids, and ground 
yellow dent corn [12]; influence of cellulose polymerization 
degree and crystallinity on kinetics of cellulose degradation 
[13]; kinetic models of xylose, glucose, arabinose, furfural, 5-

(hydroxymethyl)-2-furaldehyde and acetic acid of dilute-acid 
hydrolysis of wheat straw [8]. However, chain length 
distribution during acid hydrolysis of lignocellulosic biomass 
polysaccharide has not been explored. Actually, the hydrolysis 
of both hemicellulose and cellulose does not directly convert 
polysaccharides to sugars monomer but to hydrocellulose (a 
product with a reduced degree of polymerization) [11], [14], 
[15]. In addition, the kinetic models developed usually 
obtained from isothermal condition. According Demirbas 
(2008) [3], acid hydrolysis of lignocellulosic material can 
occur at high temperatures. In those hydrolysis, the reaction 
has in fact started during heating-up to the desired 
temperatures. So, non-isothermal kinetic model will be of 
importance. In the present work, we developed a simulation 
model of acid hydrolysis at non-isothermal condition. To be 
more detailed, the aim of this research was to model the chain 
length distribution of lignocellulosic polysaccharides during 
acid hydrolysis at non-isothermal condition, using Monte 
Carlo method. Previous study show Monte Carlo simulation 
can be applied in predicting chain length distribution of a 
number of types of chemical reactions [16]–[20]. In this study, 
the Monte Carlo simulation was developed based on the 
imagination that water molecules attack the bonds of two 
glucose units, so that new sugar polymer formed [11], as 
shown in Figure 1. In order to verify the validity of the Monte 
Carlo simulation, the results of the simulation of relatively 
short molecules were compared to the ones obtained from 
kinetic models (simultaneous differential equations solved by 
Runge-Kutta method). The Monte Carlo simulation was then 
applied to the hydrolysis of long molecules, where kinetic 
modeling via simultaneous differential equations becomes very 
complicated. 

II. METHODS OF SIMULATION  

A. Kinetic of acid hydrolysis of lignocellulosic material 
The structure of lignocellulosic material is composed of 

cellulose, hemicellulose, and lignin. Cellulose is crystalline 
glucose polymer and hemicellulose is a complex amorphous 
polymer. In acid hydrolysis of lignocellulosic material, the 
acid release protons that break the heterocyclic bonds between 
the sugar monomers in the polymeric chains. Accordingly, 
analysis of the cellulose and hemicellulose contents can be 

carried out using a quantitative acid hydrolysis under standard 
conditions [21] that stoichiometrically converts cellulose and 
hemicellulose to sugar. The sugars of the hydrolizate solutions 
produced by acid hydrolysis can be analyzed using Fehling 
method calibrated by standard glucose solution [15]. Using 
those methods, the amount of sugar obtained at various time of 
hydrolysis and initial amount of polysaccharide can be 
analyzed. The amounts are usually expressed as glucose 
equivalent. Based on those, the conversions at various times of 
hydrolysis can be calculated. Most of kinetic modeling of 
hydrolysis assumes a pseudo one step reaction, polysaccharide 
to sugars. Usually, the acid hydrolysis reaction in the 
literatures considers a pseudo-homogeneous irreversible first-
order reaction. The first model used successfully was proposed 
by Saeman [22] based on the hydrolysis of cellulose from fir 
wood using sulphuric acid catalyst. Later, the model was also 
applied to the hydrolysis of the hemicellulosic fraction [8], 
[23]. 

In this study, a more realistic model was applied. The acid 
hydrolysis of lignocellulosic material may be simplified as a 
consecutive depolymerization of glucose polymers as follows: 

 

 (1) 
 (2) 
 (3) 

 . 
 .  

 (4) 
 

in which i is the degree of polymerization of the polymer. 
It obvious that polymers A(j) are formed as the products of 

hydrolysis of longer polymers, and simultaneously degraded to 
shorter polymers. Assuming homogeneous reaction and the 
reaction is first order to the polysaccharide (the water is in 
excess) as well as applying material balances in batch reaction, 
the differential equation for polymer A(j) can be formulated as 
follows: 

 

  (5) 
 
In which Aj = concentration of polymer A(j), k = reaction rate 
constant (assumed to be the same for all chain length), j = 
1,2,3,…, NINIT, in which NINIT is the degree of polymerization 
of initial polysaccharide. Hence, the kinetics modeling results 
in NINIT simultaneous differential equations. For large NINIT 
(long polysaccharide) the equations became complicated. 
Theoretically, the degree of polymerization of cellulose 
depends on the type of plants, and is estimated to be from 400 
to 27000 glucan units [13], [15], [24]. In this simulation, the 
reaction rate constant is influenced by temperature and was 
expressed by Arrhenius equation. In the Arrhenius equation, 
the activation energy of hydrolysis of lignocellulosic material 
of forest and agricultural biomass with 12.04 M sulfuric acid 
as catalyst are about 76.6 and 68.9 kJ/mol, respectively [25]. 
Meanwhile, the empirical equation of temperature as function 

of time can be formulated based on the experimental data. 
The set of equations can be solved numerically, for example 

by Runge-Kutta method [26], so the concentration of each 
polymers and monomer at various times can be obtained. In 
this simulation, the values of initial polymer concentration 
(C0), NINIT, pre-exponential factor (A), activation energy (E), 
reaction time (∆t) were taken to be 1 mol/L, 80, 6.106 min-1, 80 
kJ/mol, and 0.1 min, respectively, and the time dependence of 
the temperature is expressed as follow: 

 

  (6) 
 

Based on the Runge-Kutta results (concentrations of 
monomer and polymers at various times), the total mol of 
depolymerization products at certain time (Ctotal), the the ratio 
of Cpoly to C0, and the chain length distribution population (the 
ratio of Cpoly to Ctotal) can be then calculated. 

B. Monte Carlo simulation 
In this paper, the Monte Carlo simulation was applied to the 

acid hydrolysis of lignocellulose polysaccharide. The 
algorithm was designed not only to simulate the kinetic 
behavior of depolymerization but also to obtain the chain 
length distribution. Imagination of the acid hydrolysis 
mechanism of lignocellulosic biomass as the basis of the 
Monte Carlo simulation is schematically shown in Figure 1. 

 

 
Fig. 1. Attacks of hydronium ion on cellulose molecule [11]. 

 

As mentioned previously, the reaction is imagined occurs 
between the inter-glucose bond and water molecule. Since the 
amount of water is in excess, the reaction is first order to the 
inter-glucose bond. It means that if the number of the bond is 
NBOND, the number of reaction per unit of time should be 
equals to k.NBOND.  Hence the average time interval between 
reactions is: 

 

 (7) 

Supposed that the initial number of the bonds is NBOND, the 
initial average time interval follows equation (7). If the number 
of bonds at certain time is N, the probability of reaction in the 
interval ∆t is: 

 

 (8) 

For non-isothermal condition, the correction due to the 
change of k value needs to be incorporated. If the maximum 
value of maximum temperature is Tmax, with the corresponding 
value of k be kmax, and the number of bonds is NBOND, then the 
average time interval between reactions (∆tR) can be applied 

International Journal of Chemical, Environmental & Biological Sciences (IJCEBS) Volume 1, Issue 3 (2013)  ISSN 2320-4079;  EISSN 2320–4087 

508



by equation (9). 
 

 (9) 

If the time interval chosen is ∆t, the number of bonds is N 
and the temperature is T, with the corresponding value of 
reaction rate constant k, the probability of reaction in the time 
interval of ∆t is: 

 

 (10) 

The Monte Carlo simulation can be then developed based 
on equations (9) and (10). The simulation started with number 
of polysaccharide molecules of Nmol0, and the degrees of 
polymerization of all initial molecules NINIT. The number of 
bonds can be easily calculated as: 

 

 (11) 

The bonds are assigned indexes I = 1, 2, 3,..., NBOND. All of 
the bonds are then assigned status, 1 if exist and 0 if does not 
exist. Hence, all initial statuses are 1, except bonds of indexes 
of NINIT, 2NINIT, 3NINIT,…, NBOND (status = 0). The ratio of 
N/NBOND can be simulated by generating random integer 
number in the interval of 1 to NBOND, say i. If bond index i 
exist (status = 1), reaction occurs, if does not exist (status = 0), 
the reaction does not occur. If bond index i reacts, its status is 
changed to 0. The change of status during time interval ∆t can 
be simulated by the following steps: 

1. Calculated k at the corresponding temperature. 
2. Calculated y = (∆t/∆tR)(k/kmax). 
3. Generated random number γ' in the interval of 0 to 1. 
4. If γ’ less or equal to y, forward to step 5, else reaction 

does not occur, so no status change. 
5. Generated random number γ’’ in the interval of 0 to 1.  
6. Generated random integer number i in the range of 1 to 

NBOND using the formula: i = 1 + γ’’ (NBOND – 1) 
7. If bond index exist (status = 1), reaction occurs and the 

status of the bond be changed to 0, else (status = 0) 
reaction does not occur. 

Starting from t = 0, the statuses of all of the bonds at various 
times can be simulated. The degree of polymerization of each 
molecule can be then known as the number of integers between 
two non-existing bonds. The number of molecules with certain 
degree of polymerization (Ni) can be then counted. The ratio 
of the concentration of the molecule with degree of 
polymerization i to the concentration of the initial 
polysaccharide can be approximated as Ni/NBOND. The 
program to conduct the Monte Carlo simulation is presented as 
follows. 

 
 
 
 

C     BATCH HYDROLISIS 
C     NON-ISOTHERMIC 
C     MONTE CARLO METHOD 
C     A(J) + H2O ------> A(1) + A(J-1) 
C     A(J) + H20 ------> A(2) + A(J-2)  
C     A(J) + H20 ------> A(3) + A(J-3)  
C            . 
C            . 
C            .  
C     A(J) + H20 ------> A(J-1) + A(1)   
C     MAY 20, 2012  
      DIMENSION ISTAT(500000),FMOL(50000) 
C     DEGREES OF POLYMERIZATION OF INITIAL POLYMERS 
      NINIT=80 
C     MAXIMUM (FINAL)TEMPERATURE OF REACTION, K 
      TMAX=493.       
C     MAXIMUM REACTION RATE CONSTANT, MOL/L/MIN 
      TEMP=TMAX 
      CALL AKREAC(TEMP,AK) 
      AKMAX=AK 
C     TIME OF SIMULATION, MIN 
      TEND=40. 
C     NUMBER OF INITIAL POLYMER 
C     (NMOL0*NINIT MUST BE LESS THAN DIMENSION OF ISTAT)       
      NMOL0=500 
C     NUMBER OF INITIAL BONDS       
      NBOND=NINIT*NMOL0       
C     MINIMUM REACTION TIME AT INITIAL CONCENTRATION, MIN 
      DELTR=1./AKMAX/FLOAT(NBOND) 
C     TIME INCREMENT (DELT MUST BE LESS OR EQUAL TO DELTR) 
      DELT=0.001 
      IF(DELT .LE. DELTR) GOTO 15 
      WRITE(*,10) 
   10 FORMAT(2X,'DELT TOO LARGE, USE SMALLER DELT!')    
      GOTO 290 
   15 IF(DELT .GE. 0.2*DELTR) GOTO 20 
      WRITE(*,18) 
   18 FORMAT(2X,'DELT TOO SMALL, USE LARGE DELT!') 
      GOTO 290     
C     PROBABILITY OF REACTION 
   20 PROB=DELT/DELTR                   
C     NUMBER OF SIMULATION STEPS       
      NSIM=INT(TEND/DELT+0.001) 
C     TIME INCREMENT FOR EVALUATION 
      DELTP=5.       
      NEVAL=INT(DELTP/DELT+0.001) 
C     INITIAL STATUS OF BONDS(1=EXISTED, 0=NOT EXISTED) 
      DO 30, I=1,NBOND 
      ISTAT(I)=1 
   30 CONTINUE 
      DO 40,I=NINIT,NBOND,NINIT 
      ISTAT(I)=0 
   40 CONTINUE       
C     SIMULATION STARTS HERE 
      ISIM=0 
      IEVAL=0 
C     INDEX OF REACTING BOND         
   50 TIME=FLOAT(ISIM)*DELT 
      TEMP=413.+5.5*TIME**0.91 
      IF(TEMP .GT. TMAX)THEN 
      TEMP=TMAX 
      ENDIF    
      CALL AKREAC(TEMP,AK) 
      RATIO=AK/AKMAX 
      GAMA=RANDOM() 
      Y=PROB*RATIO 
      IF(GAMA .GT. Y) GOTO 60  
      IREAC=1+RANDOM()*FLOAT(NBOND-1) 
      IF(ISTAT(IREAC) .EQ. 0) GOTO 60 
      ISTAT(IREAC)=0 
   60 ISIM=ISIM+1 
      IEVAL=IEVAL+1 
      IF(IEVAL .EQ. NEVAL) GOTO 90 
      GOTO 50 
   90 CALL POPULATION(NINIT,NMOL0,NBOND,ISTAT,FMOL)    
      WRITE(*,250) TIME 
      WRITE(*,255) 
      WRITE(*,260) 
      WRITE(*,270) 
      TOTAL=0. 
      DO 120,I=1,NINIT 
      TOTAL=TOTAL+FMOL(I) 
  120 CONTINUE     
      DO 130,I=1,10 
      WRITE(*,280)I,FMOL(I),FMOL(I)/TOTAL 
  130 CONTINUE 
      WRITE(*,270) 
      IEVAL=0 
      IF(ISIM .LT. NSIM) GOTO 50                         
  250 FORMAT(/2X,'TIME = ',F7.2,' MIN') 
  255 FORMAT(2X,'------------------') 
  260 FORMAT(4X,'NPOLY',4X,' CI/C0  ',4X,' POP(I) ') 

  270 FORMAT(4X,'-----',4X,'--------',4X,'--------') 
  280 FORMAT(5X,I4,3X,F9.3,6X,F6.4)           
  290 STOP 
      END  
 
      SUBROUTINE POPULATION(NINIT,NMOL0,NBOND,ISTAT,FMOL) 
C     CALCULATING DISTRIBUTION BASED ON COUNTED NUMBER DATA 
      DIMENSION ISTAT(500000),JSUM(20000),FMOL(20000) 
      DO 640,I=1,NINIT 
      JSUM(I)=0 
  640 CONTINUE 
      ISTART=0  
      DO 700,I=1,NBOND 
      IF(ISTAT(I).EQ. 1) GOTO 700    
      IEND=I 
      NPOLY=IEND-ISTART 
      JSUM(NPOLY)=JSUM(NPOLY)+1 
      ISTART=IEND 
  700 CONTINUE 
      DO 720,I=1,NINIT 
      FMOL(I)=FLOAT(JSUM(I))/NMOL0      
  720 CONTINUE   
      RETURN 
      END 
 
      SUBROUTINE AKREAC(TEMP,AK) 
      AK=6.E6*EXP(-80000/8.314/TEMP)             
      RETURN 
      END                                         

III. RESULTS AND DISCUSSION 
The results of Monte Carlo simulation are compared to the 

ones of differential equation approach solved by Runge-Kutta 
method, for relatively short chain initial molecules (NINIT = 
80), in which differential equation approach is still feasible. 
The simulation run at initial concentration, C0 = 1 mol/L, 
Arrhenius parameters of pre-exponential factor = 6.106 min-1, 
and activation energy = 80 kJ/mol, temperature formula as a 
function of time (min), T (K) = 413 + 5.5t0.91. The Monte 
Carlo simulation used Nmol0 = 500. The results are shown in 
figures 2 and 3. From those figures, it can be seen that the 
Monte Carlo simulation produces close approximates 
compared to the differential equation approach. It was 
observed also that differential equation approach was not 
feasible for relatively long chain initial molecules (degree of 
polymerization of more than 1000), since it took relatively 
long time to be conducted in an ordinary personal computers. 
Meanwhile, the Monte Carlo simulation can be still feasible 
for relatively long chain molecule (tried to be run up to NINIT = 
100000). 

 

 
Fig. 2. The chain length distribution of acid hydrolysis of 

lignocellulosic material by the kinetic model via simultaneous 
differential equations (spotted line) as well as the Monte Carlo 

method (solid line) at reaction time of 35 min. 
 

 
 

Fig. 3. The chain length distribution of acid hydrolysis of 
lignocellulosic material by the kinetic model via simultaneous 
differential equations (spotted line) as well as the Monte Carlo 

method (solid line) at reaction time of 40 min. 
 

The results of Monte Carlo simulation for NINIT = 1000, 
Nmol0 = 500, initial concentration, C0 = 1 mol/L, Arrhenius 
parameters of pre-exponential factor = 6.106 min-1 and 
activation energy = 80 kJ/mol, temperature formula as a 
function of time (min), T (K) = 413 + 5.5t0.91 are presented in 
figure 4. It can be seen that the Monte Carlo simulation results 
at relatively long chain initial molecules are conceivable. 

The conversion to sugars can be obtained easily computed 
from the simulation results. For example, if the laboratory 
chemical analysis detects the amount of sugars in the form of 
monosaccharides and disaccharides expressed as glucose 
equivalent (say N1 moles), as well as the initial amount 
polysaccharides (say N0 moles), the conversion is usually 
calculated as x = N1/N0. The simulated conversion can be 
easily calculated as: x = (number of monosaccharide+2 
(number of disaccharide)/NBOND. Comparison of the Monte 
Carlo simulation results and the relevant experimental data can 
be used to evaluate the more realistic kinetics model and their 
more fundamental parameter. The currently available 
parameters are usually obtained from simple kinetics models 
assuming pseudo direct reaction from polysaccharide to 
sugars. For example, the research of dilute-acid hydrolysis of 
Lantana camara with condition of concentration acid as 
catalyst of 1% v/v, reaction time of 45 min, and temperature of 
140 oC, the sugar yield obtained of 16,29%. Likewise, for 
biomass of olive tree with condition of concentration acid as 
catalyst of 1% v/v, reaction time of 10 min, and temperature of 
180 oC, the sugar yield obtained of 26,23% [27]. This sugar 
yield number calculated with the Monte Carlo method is about 
16% and 26% for Lantana camara and olive tree biomass, 
respectively, with the values of degree of polymerization of 
400-825 [24] and the activation energy of 80 kJ/mol [25]. The 
more realistic models are expected to enable deeper 
understanding of the reaction as well as to give more accurate 
quantitative information to support the design of equipment for 
industrial scale. Monte Carlo simulation can also be modified 
to more complicated mechanisms of hydrolysis, for example 
by considering degradation of sugars, as well as the present of 
several types of sugars. 

Fig. 4. The chain length distribution of acid hydrolysis of 

lignocellulosic material at non-isothermal condition by the Monte 
Carlo method at initial degree of polymerization (NINIT)=1000, initial 

number of molecules (Nmol0)=500, at various times of reaction. 

IV. CONCLUSIONS 
The Monte Carlo simulation successfully simulated to 

predict chain length distribution of polysaccharides in acid 
hydrolysis of lignocellulosic biomass for ethanol production. 
For initially low degree of polymerization polysaccharide, in 
which kinetics modeling via simultaneous differential 
equations is still feasible, the Monte Carlo simulation results 
are in good agreement with the ones from kinetics model via 
simultaneous differential equations, as well as with the 
experimental data obtained by another researcher. Likewise, at 
initially high degree of polymerization polysaccharide (tried 
up to 100000), the Monte Carlo simulation is still applicable, 
whereas the simultaneous differential equation approach is not 
feasible. 

